


DE LA RECHERCHE À L'INDUSTRIE



DÉBUTER AVEC CAST3M

CALCULS THERMO MÉCANIQUES

DISPONIBLE SUR : [HTTP://WWW-CAST3M.CEA.FR/INDEX.PHP?XML=FORMATIONS](http://www-cast3m.cea.fr/index.php?xml=formations)



François DI PAOLA

www.cea.fr


DERNIÈRE MODIFICATION : 17 MARS 2022

- **Présentation de Cast3M**
- **Le langage Gibiane**
- **Travaux dirigés**
comportement thermo-mécanique d'une structure
avec cavité circulaire
- **Compléments**
- **Description des objets Gibiane**

PRÉSENTATION DE CAST3M

Logiciel de simulation utilisant la méthode des **éléments finis** en **mécanique/thermique** des **structures** et des **fluides**


- Résolution d'**équations aux dérivées partielles** par la méthode des éléments finis
- **Systeme complet** : solveur, pré/post-processeur, visualisation, import/export des données ...




- Basé sur un langage de commande : **Gibiane** (orienté objet)

DE NOMBREUX DOMAINES D'APPLICATION


- **Mécanique des structures (historique)**
Quasi-statique (non linéarités matériau, géométrie, conditions limites)
Contact/frottement, **Flambage**
Dynamique (temporelle, modale, interaction fluide structure)
Rupture (XFEM, propagation dynamique, zones cohésives, ...)



- **Thermique**
 Conduction, convection, rayonnement



- **Fluides**




- **Métallurgie**


- **Magnétostatique**

- **Diffusion multi espèces (**

- **Couplage thermo-hygro-mécanique**



SCAL



AMPLITUDE DEFORMEE

Temperature au temps 0.00000E+00 (0 s)

CAST3M FEKIT
 AM
 DE

Proportions de BAIN au temps : 0.00000E+00

1970

CEASEMT : plusieurs codes de calcul en mécanique

COCO (maillage), **ESPACE**, **TEMPS**, **VISU** (post-traitement),
SANSON (prop. équivalentes)

TEDEL (poutres, tuyaux), **TRICO** (coques), **BILBO** (massifs)

PASTEL (2D, plasticité) → **INCA**

DELPHINE (thermique), **AQUAMODE** (analyse modale)

1981

Démarrage de **GIBI** (mailleur)

1983

Lancement de **Castem 2000** (basé sur GIBI)

1986

Présentation officielle de Castem2000

Procédures, mécanique des fluides et autres physiques

1990



Développement d'outils métier (Toutatis, Esus, ...)

2000

Castem 2000 → **Cast3M**

Plateforme de développement d'outils métier (Pléiades, Alliance, ...)

Nouveaux outils métier (Brasero, Gerboise, Rotor, ...)



- **Multi plateformes**
Windows, Linux, macOS ...
- **Où télécharger Cast3M ?**
<http://www-cast3m.cea.fr/index.php?page=dlcastem>
- **Accès au code source**
Développement communautaire
Compilateur / Editeur de liens fournis
- **Prix**
Gratuit pour la recherche et l'enseignement
Payant pour une utilisation commerciale
- **Quelques utilisateurs/clients**
Universités, écoles d'ingénieurs ...
IRSN, EDF, SNCF, CNRS, Framatome, Air Liquide, CERN, ...

Outil de référence IRSN pour les analyse de sureté des installations nucléaires françaises
Outil de référence Framatome pour l'analyse en mécanique de la rupture

COMMENT UTILISER CAST3M ?

- 1) Écrire un fichier texte en Gibiane l'enregistrer dans un répertoire de travail

- 2) Ouvrir un terminal / invite de commandes et se placer dans le répertoire

quelques commandes Linux basiques :

ls lister le contenu d'un répertoire
cd ici/la_bas changer de répertoire
pwd afficher le répertoire courant

- 3) Lancer Cast3M sur ce fichier
castem22 hello.dgibi

On peut aussi l'utiliser sans fichier .dgibi (mode interactif) :

castem22

```

C:\Windows\System32\cmd.exe
Microsoft Windows [version 10.0.19042.1165]
(c) Microsoft Corporation. Tous droits réservés.

C:\>cd directory_1

C:\directory_1>dir
Le volume dans le lecteur C s'appelle Windows
Le numéro de série du volume est 6EE7-3688

Répertoire de C:\directory_1

15/09/2021 15:35 <DIR>      .
15/09/2021 15:35 <DIR>      ..
15/09/2021 15:31             28 hello.dgibi
15/09/2021 15:35             34 hello.trace
                2 fichier(s)          62 octets
                2 Rép(s) 130 248 232 960 octets libres

C:\directory_1>castem21 hello.dgibi
  
```

■ Le site web Cast3M : "*the place to be*"

<http://www-cast3m.cea.fr>

- *Présentation de Cast3M*
- *Formation et tutoriels vidéo*
- *Documentation (notices, manuels, sources, exemples)*
- *Fiches d'anomalie et de développement*
- *Téléchargements*
- *Contact : support Cast3M*
- *Communauté : liste de diffusion, club Cast3M*

LE LANGAGE GIBIANE

LE LANGAGE GIBIANE : PRÉSENTATION

- Langage destiné au calcul EF mais aussi un **vrai langage de programmation**

Objets classiques (entiers, flottants, chaînes, logiques, tables ...)
Instructions conditionnelles
Boucles itératives
Sous structuration
Récursivité ...



- Langage **interprété**

Le programme peut être exécuté dès que le script est modifié
Le programme peut être exécuté en mode interactif

- Langage **orienté objet**

Tout est traité comme un objet
Pas besoin de déclarer les variables ou de spécifier leur type

- Mots clefs en français

- Programmation facile et rapide

GIBIANE : RÈGLES DE SYNTAXE

■ Ligne(s) de commande

500 caractères max par instruction

Une instruction peut être écrite sur plusieurs lignes

Se termine par un point virgule ;

Le symbole d'affectation est le signe égal =

■ Insensibilité à la casse pour tous les noms d'objets

TOTO = 3.14 ;

A = 2. * tOTO ; ici la variable A vaut bien 6.28

sauf les chaînes de caractères

'blabla' ≠ 'BLABLA'

délimitées par des simples quotes

mot1 = 'Salut a vous' ;

■ Fin du fichier de données

par la commande **FIN** ;

→ arrêt de Cast3M

par une ligne vide ou un EOF

→ mode interactif

■ Ligne de commentaire : commence par *

■ Lignes vides autorisées partout

■ Pas de priorité des opérations (lecture de gauche à droite)

$1+2*3 = 9$ → penser à utiliser des parenthèses

$1+(2*3) = 7$

■ Quelques interdictions

Pas de **tabulations** → messages d'erreur incompréhensibles

Pas de **double quotes** "

■ Quelques (fortes !) recommandations

Pas de **caractères spéciaux** (é, ç, ~, et autres œ)

Utiliser une **indentation** (avec des espace) !!!

Régler son **éditeur de texte**

coloration syntaxique, remplacement des tabulations par des espaces, ...

■ Quelques erreurs classiques

Point virgule ; oublié en fin de ligne

→ la lecture de l'instruction continue !

Apostrophe ' oubliée à la fin d'une chaîne de caractères

→ la définition de la chaîne continue !

■ Pas bien !

```
N = 7 ; F = 1 ;i=1;
REPETER boucle1;
sI (i <eg N) ;
f = F * I ;I=i+1;
SINON;quitter BOUCLE1 ;
finsi;
FIN BOUCLE1;
message F ;FIN ;
```

■ Bien !

```
** Initialisation d'un entier N
N = 7 ;

** Calcul de la factorielle du nombre entier N
F = 1 ;
I = 1 ;
REPETER BOUCLE1 ;
    SI (I <EG N) ;
        F = F * I ;
        I = I + 1 ;
    SINON ;
        QUITTER BOUCLE1 ;
    FINSI ;
FIN BOUCLE1 ;

** Affichage du resultat
MESSAGE F ;

FIN ;
```

■ Définition

Désigne toute **structure de données/résultats** munie d'un **type** (éventuellement d'un sous-type) et d'un **nom**

■ Nom des objets

Donné par l'utilisateur

Limité à **24** caractères (a...z A...Z 0...9 _)

Pièges

Plus de 24 caractères : les surnuméraires sont ignorés

Utilisation du tiret – → interdit !

Caractères accentués é, è → interdit !

■ Type des objets

Il existe plus de 40 types d'objets différents

Une liste des objets les plus utilisés est donnée à la fin de cette présentation ([lien](#))

■ Exemples (non exhaustif)

```

OBJ1 = 3 ;                               type ENTIER
OBJ2 = 3.14 ;                             type FLOTTANT
OBJ3 = 'Comment ca va ?' ;               type MOT
OBJ4 = VRAI ;                             type LOGIQUE

```

```

poin1 = POIN 0. 0. ;                      type POINT
poin2 = POIN 1. 3. ;                      type POINT
OBJ5 = DROI 8 poin1 poin2 ;               type MAILLAGE

```

```
LIST OBJ5 ;
```

```
MAILLAGE 3520406 : 8 élément(S) de type SEG2
```

```
0 sous-référence(s)
```

```
1ère ligne  numéro élément : 2ème couleur : 3ème... noeud(s)
```

1	2	3	4	5	6	7	8
DEFA							
1	3	4	5	6	7	8	9
3	4	5	6	7	8	9	2

■ Définition

Désigne tout **traitement** muni d'un **nom** (instruction Gibiane) qui construit un ou plusieurs **objets nouveaux** à partir d'un ou plusieurs objets existants

■ Noms des opérateurs

Imposé à l'utilisateur

Ce sont des instructions Gibiane

Insensibles à la casse

Cast3M ne lit que les 4 premiers caractères (**DROITE = DROI**)

Quelques exceptions : forme abrégée

DROI → **D** (ou **d**)

CERC → **C** (ou **c**)

■ Exemples d'appel à un opérateur (invocation)

Cas courants (1 objet à gauche du =)

```
obj1 = OPER obj2 ;  
obj3 = OPER obj4 obj5 ;  
obj6 = obj7 OPER obj8 obj9 ;
```

Cas exceptionnels (plusieurs objets à gauche du =)

```
obj1 obj2 obj3 = OPER obj4 obj5 ;
```

Opérateur "**sans nom**" : création de POINTS

en dimension 2 `Point1 = 0. 0. ;`

en dimension 3 `Point1 = 0. 0. 0. ;`

■ L'ordre des opérandes

est **indifférent** si les opérandes sont de **type différents**

(sauf exception dans la documentation)

est **important** si plusieurs opérandes du **même type**

■ Surcharge d'un objet

Toujours possible, l'ancien objet disparaît

`A = 'Salut' ;` → A est du type MOT

`B = 28 ;`

`C = 3 ;`

`A = B**C ;` → A est du type ENTIER et vaut 21952

■ Pièges

Nom d'objet = nom d'opérateur →

appel à l'opérateur impossible, sauf si on l'appelle en capitales entre quotes `A = 'OPER' B C ;` et en majuscules !

Objet nommé c, C, d ou D !

■ Définition

Commande sans symbole d'affectation =
Ne crée pas de nouvel objet

■ Exemples

```
OPTI 'DIME' 3 'ELEM' 'CUB8' ;  
OPTI 'TITR' 'Maillage de la piece' ;  
  
DEPL mail1 'PLUS' (28. -0.3 20.03) ;
```

La directive **OPTI** est généralement la première instruction utilisée
Elle permet de fixer les **options générales** de Cast3M.

Exemples :

dimension de l'espace, éléments de maillage utilisés, taille de maille,
nom du fichier de sauvegarde, calcul axisymétrique, et bien d'autres ...

■ Définition

Ensemble nommé de commandes Gibiane muni d'une liste d'opérandes d'entrée et de sortie

Analogue à une subroutine Fortran ou à une fonction C

■ Nom des procédures

Comme un objet ordinaire (une procédure est un objet de type PROCEDUR)

■ Déclaration

```
DEBP ma_proc arg_e1*entier arg_e2*flottant ... arg_en*mcham1 ;  
  commande 1 ;  
  commande 2 ;  
  ...  
  commande k ;  
FINP arg_s1 arg_s2 ... arg_sm ;
```

■ Invocation

Comme un opérateur ou une directive ordinaire

```
obj1 obj2 ... objm = ma_proc ent1 flot2 ... champn ;
```

■ Il existe des procédures pré-cablées dans Cast3M

Voir la liste : <http://www-cast3m.cea.fr/index.php?page=notices>

PASAPAS	→ calculs non linéaires
FLAMBAGE	→ calculs de flambage
DYNAMIC	→ calculs dynamiques
THERMIC	→ calculs thermiques
G_THETA	→ calcul d'intégrales J et FIC (rupture)

...

il y en a d'autres, à découvrir en naviguant dans la documentation

■ Pièges

FINP manquant

→ arrêt de Cast3M, message d'erreur parfois difficile à interpréter

FINP présent mais ; manquant

→ arrêt de Cast3M, message d'erreur parfois difficile à interpréter

Invocation d'une procédure avant qu'elle ne soit déclarée

→ arrêt de Cast3M, message d'erreur de l'opérateur = parfois difficile à interpréter

■ Débugage

INFO OPER ;

→ affiche la notice d'un opérateur/directive/procédure

OPTI 'DONN' 5 ;

→ arrêt de la lecture du fichier .dgibi

→ lecture sur le terminal : **mode interactif**

OPTI 'DONN' 3 ;

→ reprise de la lecture du fichier .dgibi (là où il s'est arrêté)

LIST OBJ1 ;

→ liste le contenu de l'objet OBJ1

LIST 'RESU' OBJ1 ;

→ liste un **résumé** du contenu de l'objet OBJ1

OPTI 'DEBU' 1 ;

→ accès aux variables locales des procédures

TRAC OBJ1 (OBJ2) ;

→ trace l'objet OBJ1 (maillage, champ, déformée, ...)

MESS 'Je passe ici !' ;

→ affiche un message

■ Notices des opérateurs/directives/procédures

- 1) Directive INFO, par exemple : INFO EXTR ;
- 2) Page html locale : dans le répertoire d'installation

exemple sur un système Linux : /home/pierre_martin/CAST3M_2022/doc/index.html

exemple sur un système Windows : C:\Cast3M\PCW_22\doc\index.html

- 3) Le site web : <http://www-cast3m.cea.fr/index.php?page=notices>
attention, il s'agit de la version du jour !

■ Manuels utilisateurs

Sur le site web, à l'onglet "Documentation"


TRAVAUX DIRIGÉS

**MODÉLISATION DU COMPORTEMENT
THERMO-MÉCANIQUE
D'UNE STRUCTURE AVEC CAVITÉ**

PROBLÈME ÉTUDIÉ ET CONDITIONS AUX LIMITES

Thermique

Mécanique



■ Organisation d'un calcul élément-finis (4 grandes étapes)

1) Choix de la géométrie et du maillage

- a) Définition des points lignes, surfaces, volumes
- b) Discrétisation

2) Définition du modèle mathématique

- a) Définition des du modèle (type d'analyse, formulation, comportement matériau, types d'éléments)
- b) Définition des propriétés matérielles (module de Young, masse volumique, ...)
- c) Définition des propriétés géométriques (épaisseur des coques, moments quadratiques des poutres, ...)
- d) Définition des conditions aux limites/chargements
- f) Définition des conditions initiales

3) Résolution du problème discrétisé

- a) Calcul des matrices de rigidité et de masse pour chaque élément fini
- b) Assemblage des matrices
- c) Application des conditions limites/chargements
- e) Résolution du système d'équations

4) Analyse et post-traitement des résultats

- a) Calcul de quantités locales (déplacement, contraintes, déformation, ...)
- b) Calcul de quantités globales (déformation maximale, charge limite, ...)

■ Les fichiers solution de cette formation sont des cas tests


Téléchargeables sur le site web, à la page des exemples :
<http://www-cast3m.cea.fr/index.php?page=exemples>

Trois fichiers sont utilisés :

- *formation_debutant_1_maillage.dgibi*
- *formation_debutant_2_thermique.dgibi*
- *formation_debutant_3_mecanique.dgibi*

Objectif : créer un maillage paramétré de la structure

1. placer des points maîtres
2. mailler le contour fermé,
3. puis la surface par remplissage



Options générales et paramètres

* OPTIONS GENERALES ET TYPE D'ELEMENTS GEOMETRIQUES

OPTI 'DIME' 2 'ELEM' 'QUA8' ;

* DEFINITION DES LONGUEURS ET DENSITES DE MAILLES

LONG = 24.E-1 ;

HAUT = 10.E-1 ;

RAYON = 2.E-1 ;

NLONG = 24 ;


NHAUT = 4 ;

NBASG = 10 ;

NBASD = 10 ;

NQCG = 8 ;


NQCD = 8 ;



■ Création de points géométriques

* CREATION DES POINTS D'APPUI DU MAILLAGE

PA = 0. 0. ;
 PB = ((0.5 * LONG) - RAYON) 0. ;
 PC = (0.5 * LONG) RAYON ;
 PD = ((0.5 * LONG) + RAYON) 0. ;
 PE = LONG 0. ;
 PF = LONG HAUT ;
 PG = 0. HAUT ;
 PCEN = (0.5 * LONG) 0. ;



■ Création de lignes et du contour fermé

* CONSTRUCTION DES LIGNES DROITES (**DROI** ou **D**)

LIAB = **DROI** NBASG PA PB ;

LIDE = **DROI** NBASD PD PE ;

LIEF = **D** NHAUT PE PF ;

LIFG = **D** NLONG PF PG ;

LIGA = **D** NHAUT PG PA ;

* CONSTRUCTION DES CERCLES (**CERC** ou **C**)

(A vous de jouer: Consulter la notice)

* CONTOUR FERME OBTENU PAR ASSEMBLAGE DE LIGNES ELEMENTAIRES

CO = LIAB **ET** CE **ET** LIDE **ET** LIEF **ET** LIFG **ET** LIGA ;


■ Création de lignes et du contour fermé

* CONSTRUCTION DES LIGNES DROITES (DROI ou D)

LIAB = DROI NBASG PA PB ;
 LIDE = DROI NBASD PD PE ;
 LIEF = D NHAUT PE PF ;
 LIFG = D NLONG PF PG ;
 LIGA = D NHAUT PG PA ;

* CONSTRUCTION DES CERCLES (CERC ou C)

CE1 = CERC NQCG PB PCEN PC ;
 CE = C NQCD CE1 PCEN PD ;




* CONTOUR FERME OBTENU PAR ASSEMBLAGE DE LIGNES ELEMENTAIRES




CO = LIAB ET CE ET LIDE ET LIEF ET LIFG ET LIGA ;

■ Maillage de la surface (maillage libre depuis le contour fermé)

SU = SURF C0 ;
TRAC SU ;




Objectif : créer un maillage **régulier** de la structure



- ne contient plus que des quadrangles
- taille de maille maîtrisée
- respecte la symétrie de la pièce

Objectif : créer un maillage **régulier** de la structure

maillage réglé, symétrie, taille de maille variable




■ Maillage de la surface (maillage réglé entre 2 lignes)

NHAUT	= 10 ;	
NQCG	= 20 ;	
PH	= (<i>placer Le point PH</i>)	→ <i>POIN</i>)
LIHG	= (<i>définir Le segment HG</i>)	→ <i>DROI</i>)
LIGA	= (<i>définir Le segment GA</i>)	→ <i>DROI</i>)
CE1	= (<i>définir Le 1/4 de cercle CE1</i>)	→ <i>CERC</i>)
SU1	= (<i>mailler La surface SU1</i>)	→ <i>REGL</i>)
SU2	= (<i>faire La symétrie de SU1</i>)	→ <i>SYME</i>)
SU	= (<i>assembler Les deux surfaces</i>)	→ <i>ET</i>)

■ Maillage de la surface (maillage réglé entre 2 lignes)

NHAUT = 10 ;
 NQCG = 20 ;
 PH = (0.5 * LONG) HAUT ;
 LIHG = DROI (NQCG - NHAUT) PH PG ;
 LIGA = DROI NHAUT PG PA ;
 CE1 = CERC NQCG PB PCEN PC ;
 SU1 = REGL 'DINI' 0.05 'DFIN' 0.1 (INVE CE1) (LIHG ET LIGA) ;
 SU2 = SU1 SYME 'DROI' PCEN PH ;
 SU = SU1 ET SU2 ;

TRAC SU ;



■ Élimination des nœuds doubles, directive **ELIM**

ELIM SU 1.E-9 ;

TRAC SU ;

→ tous les nœuds de SU espacés de moins de 10^{-9} m sont fusionnés en un seul

→ ne change pas le type des éléments initiaux

→ directive à utiliser avec parcimonie !

■ Voir aussi l'opérateur **REGE**

→ régénération des éléments à nœuds confondus

→ changement du type d'éléments si besoin

■ Récupération de zones maillées

* RECUPERATION DU MAILLAGE DU DEMI CERCLE, DE LA LIGNE SUPPERIEURE

* ET DES LIGNES INFERIEURES AVEC 'POIN' ET 'ELEM'

CSU = CONT SU ;	(Récupération du contour de SU)
PCE = SU POIN 'SPHE' PCEN PC 1.E-9 ;	(Récupération des points situés sur le cercle CE)
CE = CSU ELEM 'APPU' 'STRI' PCE ;	(Récupération des éléments qui contiennent ces points)
PLHAUT = (A vous de jouer);	(Récupération des points situés sur la ligne FG)
LHAUT = (A vous de jouer);	(Récupération des éléments qui contiennent ces points)
PLBAS = (A vous de jouer);	(Récupération des points situés sur la ligne du bas)
LBAS = (A vous de jouer);	(Récupération des éléments qui contiennent ces points)

■ Récupération de zones maillées

* RECUPERATION DU MAILLAGE DU DEMI CERCLE, DE LA LIGNE SUPPERIEURE

* ET DES LIGNES INFERIEURES AVEC 'POIN' ET 'ELEM'

CSU = CONT SU ;

PCE = SU POIN 'SPHE' PCEN PC 1.E-9 ;

CE = CSU ELEM 'APPU' 'STRI' PCE ;

PLHAUT = SU POIN 'DROI' PF PG 1.E-9 ;

LHAUT = CSU ELEM 'APPU' 'STRI' PLHAUT ;

PLBAS = SU POIN 'DROI' PE PCEN 1.E-9 ;

LBAS = CSU ELEM 'APPU' 'STRI' PLBAS ;

■ Maillage de volumes (complément)

* PASSAGE EN DIMENSION 3

OPTI 'DIME' 3 'ELEM' 'CU20' ;

* VOLUME PAR TRANSLATION

VO = SU VOLU 6 'TRAN' (0. 0. 2.) ;

* VOLUME PAR ROTATION

VO = SU VOLU 10 'ROTA' 90. (0. -1. 0.) (1. -1. 0.) ;

* AFFICHAGE

TRAC VO ;



TRAC 'CACH' VO ;

OPTI 'TRAC' 'OPEN' ;

TRAC 'CACH' VO ;

* RETOUR EN DIMENSION 2

OPTI 'DIME' 2 ;



■ Sauvegarde des données et fin du programme

* NOM DU FICHIER DE SAUVEGARDE

```
OPTI 'SAUV' 'formation_debutant_1_maillage.sauv' ;
```

* ECRITURE DES FICHIERS

```
SAUV ;
```

* FIN DU PROGRAMME GIBIANE

```
FIN ;
```

- Tous les objets en mémoire sont sauvegardés
- Le fichier est binaire (format XDR)
- D'autres formats possibles (texte, ...), voir les notices de `OPTI 'SAUV'` et de `SAUV`

■ Équation de la chaleur

$$\rho c_p \frac{\partial T}{\partial t} + \operatorname{div}(-\lambda \overrightarrow{\operatorname{grad}}(T)) - q = 0 \quad \text{sur } V$$

avec :

T : température

q : source de chaleur volumique

λ : conductivité thermique

ρ : masse volumique

c_p : capacité calorifique massique

t : temps

■ Conditions aux limites

Températures imposées

$$T = T_{imp} \quad \text{sur } \partial V^T$$

Flux imposés

$$\vec{n} \cdot (\lambda \overrightarrow{\operatorname{grad}}(T)) = \varphi_{imp} + \underbrace{h(T_f - T)}_{\text{convection}} + \underbrace{\varepsilon \sigma (T_\infty^4 - T^4)}_{\text{rayonnement}} \quad \text{sur } \partial V^\varphi$$

■ Forme discrétisée (éléments finis)

$$\text{Discrétisation EF : } T(x) = [N(x)]\{T\} \quad \overrightarrow{\text{grad}}(T) = [B(x)]\{T\}$$

Formulation faible + EF :

$$[C]\{\dot{T}\} + [K]\{T\} = \{F\}$$

Matrices

$$[C] = \int_V \rho c_p [N]^T [N] dV \quad \text{matrice de capacité (J.K}^{-1}\text{)}$$

$$[K] = \int_V [B]^T [\lambda] [B] dV + \int_{\partial V \varphi} h [N]^T [N] dS \quad \text{matrice de conductivité (W.K}^{-1}\text{)}$$

Vecteur des flux nodaux équivalents (W)


$$\{F\} = \int_V [N]^T q dV + \int_{\partial V \varphi} [N]^T \left(\varphi_{imp} + hT_f + \varepsilon\sigma(T_\infty^4 - T^4) \right) dS$$

Objectif : calcul thermique *stationnaire* en températures et flux imposés
(le « temps » n'intervient pas)

$$[C]\{\dot{T}\} + [K]\{T\} = \{F\} \quad \rightarrow \text{Système linéaire}$$

1. calcul de la matrice de conductivité
2. calcul des flux de chaleur nodaux imposés
3. résolution avec RESO \rightarrow températures

$[K]$ (1^{er} membre)
 $\{F\}$ (2nd membre)
 $\{T\}$ (inconnue)



■ Restitution des objets (maillage, paramètres, ...)

* NOM DU FICHIER A RESTITUER

```
OPTI 'REST' 'formation_debutant_1_maillage.sauv' ;
```

* CHARGEMENT EN MEMOIRE

```
REST ;
```

→ Tous les objets sauvegardés sont chargés en mémoire

→ Ils sont alors accessibles de suite

* CARACTERISTIQUES DU MATERIAU

```
CONDUMAT = 210. ;
```

```
CAPAMAT = 900. ;
```

```
RHOMAT = 2700. ;
```

* TEMPERATURE INITIALE ET DE REFERENCE

```
T0 = 25. ;
```

■ Formulation mathématique

* MODELE THERMIQUE (CONDUCTION) A MATERIAU UNIFORME ET CONSTANT

MOT = MODE SU 'THERMIQUE' ;

MAT = MATE MOT 'K' CONDUMAT 'C' CAPAMAT 'RHO' RHOMAT 'TINI' T0 ;

* CALCUL DE LA MATRICE DE CONDUCTIVITE (PREMIER MEMBRE)

CON = COND MOT MAT ;

$$[K] = \int_V [B]^T [\lambda] [B] dV + \int_{\partial V \varphi} h [N]^T [N] dS$$

■ Conditions aux limites

* MATRICE DE BLOCAGE : TEMPERATURE IMPOSEE SUR LE TROU

BLT = BLOQ CE 'T' ;

* FLUX NODAUX ASSOCIES AU BLOCAGE

TMAX = 250. ;

FLT1 = DEPI BLT TMAX ;

[lien](#)

* FLUX IMPOSE SUR LA LIGNE GAUCHE

FLT2 = FLUX MOT LIGA -3.5E4 ; $\{F\} = \int_V [N]^T q dV + \int_{\partial V \varphi} [N]^T (\varphi_{imp} + hT_f + \varepsilon\sigma(T_\infty^4 - T^4)) dS$

■ Résolution du système linéaire


* CALCUL DU CHAMP DE TEMPERATURES PAR APPEL AU SOLVEUR

TCON1 = RESO (CON ET BLT) (FLT1 ET FLT2) ;

■ Visualisation des résultats

* AFFICHAGE DU CHAMP DE TEMPERATURE



TRAC TCON1 SU ;



■ Visualisation des résultats

* AFFICHAGE DU CHAMP DE TEMPERATURE

TRAC TCON1 SU CSU ;




■ Visualisation des résultats

* AFFICHAGE DU CHAMP DE TEMPERATURE


LISO1 = PROG 0. 'PAS' (TMAX / 15.) TMAX ;

TRAC TCON1 SU CSU LISO1 ;

nouvel objet LISTREEL



T
< 2.50E+02
> 2.75E+01



■ Visualisation des résultats


* AFFICHAGE DU CHAMP DE TEMPERATURE

```
TRAC TCON1 SU CSU LIS01 'TITR' '[2] Temperature' ;
```



Objectif : *calcul thermique précédent*
 + *convection*
 + *source volumique*

1. *modèle de convection*
2. *terme source*



■ Formulation mathématique

* MODELE DE CONVECTION

MOC = MODE LHAUT 'THERMIQUE' 'CONVECTION' ;

MAC = MATE MOC 'H' 100. ;

■ Matrice de conductivité (mais pour la convection !)

* PREMIER MEMBRE POUR LA CONVECTION

CONH = COND MOC MAC ; $[K] = \int_V [B]^T [\lambda] [B] dV + \int_{\partial V \varphi} h [N]^T [N] dS$

■ Vecteur flux nodal équivalent (convection)

* SECOND MEMBRE POUR LA CONVECTION

CHTC = MANU 'CHPO' LHAUT 'T' T0 ;

FLH = CONV MOC MAC CHTC ; $\{F\} = \int_V [N]^T q dV + \int_{\partial V \varphi} [N]^T (\varphi_{imp} + hT_f + \varepsilon\sigma(T_\infty^4 - T^4)) dS$

■ Vecteur flux nodal équivalent (source volumique)

* SECOND MEMBRE POUR LA SOURCE


X Y = COOR SU ;

PT1 = X POIN 'SUPERIEUR' 20.E-1 ;

PT2 = (REDU Y PT1) POIN 'INFERIEUR' 5.E-1 ;

ELSOU = SU ELEM 'APPUYE' 'STRICTEMENT' PT2 ;

FLS = SOUR (REDU MOT ELSOU) 1.E5 ELSOU ;



$$\{F\} = \int_V [N]^T q dV + \int_{\partial V \varphi} [N]^T (\varphi_{imp} + hT_f + \varepsilon\sigma(T_{\infty}^4 - T^4)) dS$$


■ Résolution du système linéaire

* CALCUL DU CHAMP DE TEMPERATURES PAR APPEL AU SOLVEUR

TCON2 = RESO (CON ET CONH ET BLT) (FLT1 ET FLT2 ET FLH ET FLS) ;

* AFFICHAGE DU CHAMP DE TEMPERATURE

TRAC TCON2 SU CSU LIS01 ;



■ Objet CHPOINT

Représente un champ de valeurs exprimées sur des POINTS (nœuds)

Exemples :

- champ scalaire de température
- champ vectoriel de déplacement (3 composantes)
- champ vectoriel de coordonnées des nœuds
- second membre d'un problème linéaire $[K]\{x\} = \{F\}$, c'est-à-dire :
 - forces nodales équivalentes
 - flux nodaux équivalents
- et bien d'autres ...

Quelques caractéristiques :

- une seule valeur possible par nœud
- ne dépend pas du maillage, seulement des nœuds !
- lors du tracé, le champ est donc continu sur le maillage

■ Objet MCHAML

Représente un champ de valeurs exprimées dans les ÉLÉMENTS d'un maillage
Exemples :

- champ de paramètres matériau
- champ de contraintes, déformations
- champ de variables internes
- gradient d'un CHPOINT
- et bien d'autres ...


Quelques caractéristiques :

- plusieurs points support possibles :
 - . points d'intégration des contraintes
 - . point d'intégration de la rigidité
 - . points d'intégration de la masse
 - . centre de gravité
 - . nœuds
- interpolé par les fonctions d'interpolation $[N]$ du modèle
- non continu d'un élément à l'autre

Objectif : *calcul thermique précédent*
 + *en transitoire (avec une température initiale = 25 °C)*

$$[C]\{\dot{T}\} + [K]\{T\} = \{F\}$$

1. *description temporelle du chargement*
2. *conditions initiales*
3. *résolution avec la procédure PASAPAS*



■ Intervalle de temps simulé

* INSTANT FINAL DU CALCUL THERMIQUE
TPSFIN = 5.E4 ;

■ Définition de chargements (CL dans l'espace et le temps)


* CHARGEMENT TEMPERATURE IMPOSÉE
LIST1 = PROG 0. (0.1 * TPSFIN) TPSFIN ;
LIST2 = PROG (T0 / TMAX) 1. 1. ;
EVT1 = EVOL 'MANU' LIST1 LIST2 ;
CHATIMP = CHAR 'TIMP' FLT1 EVT1 ;

* CHARGEMENT FLUX IMPOSÉ
LIST3 = PROG 0. (0.2 * TPSFIN) (0.4 * TPSFIN) TPSFIN ;
LIST4 = PROG 0. 0. 1. 1. ;
EVT2 = EVOL 'MANU' LIST3 LIST4 ;
CHAFIMP = CHAR 'Q' FLT2 EVT2 ;

* CHARGEMENT CONVECTION
LIST5 = PROG 0. TPSFIN ;
LIST6 = PROG 1. 1. ;
EVCTE = EVOL 'MANU' LIST5 LIST6 ;
CHACONV = CHAR 'TECO' CHTC EVCTE ;

* CHARGEMENT SOURCE VOLUMIQUE
CHASOUR = CHAR 'Q' FLS EVT2 ;

CHAT = CHATIMP ET CHAFIMP ET CHACONV ET CHASOUR ;



■ Construction de la table pour la procédure PASAPAS

* DEFINITION DE LA TABLE D'ARGUMENTS A FOURNIR EN DONNEE D'ENTREE A LA
* PROCEDURE PASAPAS

```
TAB1                = TABL ;
TAB1 . 'MODELE'     = MOT ET MOC ;
TAB1 . 'CARACTERISTIQUES' = MAT ET MAC ;
TAB1 . 'BLOCAGES_THERMIQUES' = BLT ;
TAB1 . 'CHARGEMENT' = CHAT ;
TAB1 . 'TEMPS_CALCULES' = PROG 0. 'PAS' (0.02 * TPSFIN) TPSFIN ;
```

■ Résolution avec la procédure PASAPAS


* APPEL A PASAPAS
PASAPAS TAB1 ;

■ Post traitement : courbes d'évolution, tracés de champs

- * EVOLUTION TEMPORELLE DE LE TEMPERATURE EN DEUX POINTS POUR VERIFIER
- * QUE L'ON A ATTEINT L'ETAT STATIONNAIRE

```
PMIL      = SU POIN 'PROC' ((0.5 * LONG) (0.5 * HAUT)) ;
EV1       = EVOL 'ROUG' 'TEMP' TAB1 'TEMPERATURES' 'T' PMIL ;
EV2       = EVOL 'BLEU' 'TEMP' TAB1 'TEMPERATURES' 'T' PG ;
```

DESS (EV1 ET EV2) ;



■ Post traitement : boucle itérative pour le tracé




* BOUCLE SUR TOUS LES PAS DE TEMPS ET TRACE DU CHAMP DE TEMPERATURES

```

N1      = DIME (TAB1 . 'TEMPERATURES') ;
REPE B1 N1 ;
  T_I    = TAB1 . 'TEMPERATURES' . (&B1 - 1) ;
  TPS_I  = TAB1 . 'TEMPS' . (&B1 - 1) ;
  PRC_I  = ENTI (100. * TPS_I / TPSFIN) ;
  MOT_I  = CHAI '[3] Temperature au temps ' TPS_I ' (' PRC_I ' %)' ;
  TRAC T_I SU CSU 'TITR' MOT_I LIS01 ;
FIN B1 ;

```

Compteur de boucle
1, 2, 3, ... , N1



■ Création d'une procédure (calcul du vecteur flux de chaleur)

```
* CREATION D'UNE PROCEDURE POUR REPRESENTER LE "FLUX DE CHALEUR"
DEBP VECFLU CHP1*'CHPOINT' MOD1*'MMODEL' MAT1*'MCHAML' ;
* CALCUL DU GRADIENT DE T ET CHANGEMENT DU TYPE
G1      = GRAD CHP1 MOD1 ;
G2      = CHAN 'TYPE' G1 'CARACTERISTIQUES' ;
* MULTIPLICATION DES CHAMPS ENTRE EUX
Q       = MAT1 * G2 (MOTS 'K' 'K') (MOTS 'T,X' 'T,Y') (MOTS 'QX' 'QY') ;
Q       = -1. * Q ;
* CREATION D'UN OBJET VECTEUR
VEC1    = VECT Q MOD1 (MOTS 'QX' 'QY') 2.E-6 ;
FINP VEC1 ;
```

■ Post traitement : tracés du vecteur flux de chaleur

* BOUCLE SUR TOUS LES PAS DE TEMPS ET TRACE VECTEUR FLUX DE CHALEUR


```
REPE B1 N1 ;
```

```
  T_I      = TAB1 . 'TEMPERATURES' . (&B1 - 1) ;
```

```
  VF_I     = VECFLU T_I MOT MAT ;
```

```
  TRAC VF_I CSU ;
```

```
FIN B1 ;
```




■ Post traitement : tracés du vecteur flux de chaleur et des lignes d'isovaleurs

* VECTEUR FLUX ET CHAMP DE TEMPERATURE SOUS FORME DE LIGNES D'ISOVALEURS

OPTI 'ISOV' 'LIGN' ;


TRAC VF_I T_I SU CSU 15 ;

OPTI 'ISOV' 'SURF' ;



Objectif : calcul thermique précédent
+ rayonnement

1. ajout modèle et chargement de rayonnement



■ Modèle thermique de rayonnement à l'infini

* MODELE DE RAYONNEMENT

MOR = MODE LHAUT 'THERMIQUE' 'RAYONNEMENT' 'INFINI' ;
MAR = MATE MOR 'EMIS' 0.8 ;

■ Chargement de température de rayonnement

* CHARGEMENT RAYONNEMENT

CHTR = MANU 'CHPO' LHAUT 1 'T' T0 ;
CHARAY = CHAR 'TERA' CHTR EVCTE ;
CHAT = CHAT ET CHARAY ;

■ Construction de la table pour la procédure PASAPAS

* REDEFINITION DE LA TABLE TAB1 POUR PASAPAS

```
TAB1 = TABL ;  
TAB1 . 'MODELE' = (A vous de jouer)  
TAB1 . 'CARACTERISTIQUES' = (A vous de jouer)  
TAB1 . 'BLOCAGES_THERMIQUES' = (A vous de jouer)  
TAB1 . 'CHARGEMENT' = (A vous de jouer)  
TAB1 . 'TEMPS_CALCULES' = (A vous de jouer)  
Autres indices ?? (A vous de jouer)
```

■ Résolution avec la procédure PASAPAS

* APPEL A PASAPAS
(A vous de jouer)

■ Construction de la table pour la procédure PASAPAS

* REDEFINITION DE LA TABLE TAB1 POUR PASAPAS

```
TAB1 = TABL ;
TAB1 . 'MODELE' = MOT ET MOC ET MOR ;
TAB1 . 'CARACTERISTIQUES' = MAT ET MAC ET MAR ;
TAB1 . 'BLOCAGES_THERMIQUES' = BLT ;
TAB1 . 'CHARGEMENT' = CHAT ;
TAB1 . 'TEMPS_CALCULES' = PROG 0. 'PAS' (0.02 * TPSFIN) TPSFIN ;
TAB1 . 'CELSIUS' = VRAI ;
```

■ Résolution avec la procédure PASAPAS

* APPEL A PASAPAS

```
PASAPAS TAB1 ;
```

■ Post traitement

* EVOLUTION TEMPORELLE DE LA TEMPERATURE

EV11 = EVOL 'ROUG' 'TEMP' TAB1 'TEMPERATURES' 'T' PMIL ;

EV22 = EVOL 'BLEU' 'TEMP' TAB1 'TEMPERATURES' 'T' PG ;

TL = TABL ;

TL . 1 = 'TIRR' ;

TL . 2 = 'TIRR' ;

TL . 'TITRE' = TABL ;


TL . 'TITRE' . 1 = 'PMIL' ;

TL . 'TITRE' . 2 = 'PG' ;

TL . 'TITRE' . 3 = 'PMIL avec rayonnement' ;

TL . 'TITRE' . 4 = 'PG avec rayonnement' ;

DESS (EV1 ET EV2 ET EV11 ET EV22) 'LEGE' TL ;



■ Sauvegarde des données et fin du programme

```
OPTI 'SAUV' 'formation_debutant_2_thermique.sauv' ;
```

```
SAUV ;
```

```
FIN ;
```

■ Équation d'équilibre (statique)

$$\operatorname{div}(\boldsymbol{\sigma}) + \vec{f} = \vec{0} \quad \text{sur } V$$

■ Conditions aux limites

Déplacements imposés : $\vec{u} = \vec{d}$ sur ∂V^d

Efforts imposés : $\boldsymbol{\sigma} \cdot \vec{n} = \vec{t}$ sur ∂V^t

avec :

\vec{u} : vecteur déplacement

$\boldsymbol{\sigma}$: tenseur des contraintes

\vec{f} : forces volumiques

\vec{n} : normale à la surface

■ Discrétisation EF

$$\vec{u}(x) = [N(x)]\{u\}$$

■ Formulation faible de l'équilibre + discrétisation EF

$$\{F\} - \int_V [B]^T \{\sigma\} dV = \{0\}$$

$$\underbrace{\int_{\partial V^t} [N]^T \{t\} dS}_{\{F\}^S} + \underbrace{\int_{\partial V^d} [N]^T \{\sigma \cdot n\} dS}_{\{F\}^R} + \underbrace{\int_V [N]^T \{f\} dV}_{\{F\}^V} - \underbrace{\int_V [B]^T \{\sigma\} dV}_{[B]\{\sigma\}} = \{0\}$$

Forces nodales équivalentes (N) aux :

$\{F\}^S$ densité surfacique d'efforts imposés \vec{t}

$\{F\}^R$ densité surfacique d'efforts de réaction aux déplacements imposés \vec{d}

$\{F\}^V$ densité volumique d'efforts imposés f

$[B]\{\sigma\}$ densité volumique d'efforts intérieurs

Matrices

$[N]$ matrice des fonctions de forme (.)

$[B]$ matrice des dérivées des fonctions de forme (m⁻¹)

■ Discrétisation EF

$$\vec{u}(x) = [N(x)]\{u\}$$

■ Formulation faible de l'équilibre + discrétisation EF

$$\{F\} - \int_V [B]^T \{\sigma\} dV = \{0\}$$

■ Hypothèse petites déformations $\varepsilon(x) = [B(x)]\{u\}$ + élasticité

$$\int_V [B]^T \{\sigma\} dV = \underbrace{\int_V [B]^T [C] [B] dV}_{[K]} \cdot \{u\}$$

$$[K]\{u\} = \{F\}$$

Matrices


$[K]$ matrice de rigidité (N. m⁻¹)

Objectif :

calcul mécanique élastique
avec blocages des déplacements,
densités surfacique et volumique d'efforts imposés

$$[K]\{u\} = \{F\} \quad \rightarrow \text{Système linéaire}$$

- | | |
|--|----------------------|
| 1. calcul de la matrice de raideur | $[K]$ (1er membre) |
| 2. calcul des efforts nodaux imposés | $\{F\}$ (2nd membre) |
| 3. résolution avec RESO \rightarrow déplacements | $\{u\}$ (inconnue) |



■ Restitution des données des précédents calculs

```
OPTI 'REST' 'formation_debutant_2_thermique.sauv' ;  
REST ;
```

* CARACTERISTIQUES MECANIQUES DU MATERIAU

```
YOUNGMAT = 30.E9 ;  
NUMAT    = 0.2 ;  
ALPHAMAT = 12.E-6 ;  
SIGYMAT  = 120.E6 ;
```

■ Hypothèse des déformations planes

```
OPTI 'MODE' 'PLAN' 'DEFO' ;
```

■ Modèle mécanique élastique linéaire isotrope

```
MOM1      = MODE SU 'MECANIQUE' 'ELASTIQUE' ;  
MAM1      = MATE MOM1 'YOUN' YOUNGMAT 'NU' NUMAT 'ALPH' ALPHAMAT 'TREF' T0 'TALP' T0 ;
```

■ Matrice de raideur ou rigidité (1^{er} membre)

RI = RIGI MOM1 MAM1 ;

$$[K] = \int_V [B]^T [C] [B] dV$$

■ Conditions aux limites de déplacements imposés (1^{er} membre)

BLMX = BLOQ PA 'UX' ; (blocage selon x du point A)

BLMY = BLOQ (A vous de jouer); (blocage selon y de la ligne du bas)

RITOT = (A vous de jouer); (assemblage des rigidités)

■ Forces nodales représentatives de la pression (2nd membre)

FS = PRES (A vous de jouer);

■ Forces nodales représentatives de la gravité (2nd membre)

FV = CNEQ (A vous de jouer);

■ Résolution du système linéaire

U5 = (A vous de jouer);

■ Matrice de raideur ou rigidité (1^{er} membre)

RI = RIGI MOM1 MAM1 ;

$$[K] = \int_V [B]^T [C] [B] dV$$

■ Conditions aux limites de déplacements imposés (1^{er} membre)

BLMX = BLOQ PA 'UX' ;

BLMY = BLOQ LBAS 'UY' ;

RITOT = RI ET BLMX ET BLMY ;

■ Forces nodales représentatives de la pression (2nd membre)

FS = PRES 'MASS' MOM1 LHAUT 30.E6 ;

$$\{F\}^S = \int_{\partial V^t} [N]^T \{t\} dS$$

■ Forces nodales représentatives de la gravité (2nd membre)

ROG = MANU 'CHPO' SU 2 'FX' 0. 'FY' (-9.81 * RHOMAT) ;

FV = CNEQ MOM1 ROG ;

$$\{F\}^V = \int_V [N]^T \{f\} dV$$

■ Résolution du système linéaire

U5 = RESO RITOT (FS ET FV) ;

■ Post traitement : déformations, contraintes, maillage déformé

* TRACE DU MAILLAGE DEFORME


```
DEF_5 = DEFO SU U5 150. 'ROUG' ;
```

```
DEF_INI = DEFO SU U5 0. ;
```

```
TRAC (DEF_INI ET DEF_5) ;
```

* TRACE DU CONTOUR DEFORME


(A vous de jouer)



■ Post traitement : déformations, contraintes, maillage déformé


* TRACE DU MAILLAGE DEFORME

```
DEF_5      = DEFO SU U5 150. 'ROUG' ;
DEF_INI    = DEFO SU U5 0. ;
TRAC (DEF_INI ET DEF_5) ;
```



* TRACE DU CONTOUR DEFORME

```
DEF_5C     = DEFO CSU U5 150. 'ROUG' ;
DEF_INIC   = DEFO CSU U5 0. ;
TRAC (DEF_INIC ET DEF_5C) ;
```



■ Post traitement : déformations, contraintes, maillage déformé

En deux coups :

* CALCUL DES DEFORMATIONS

DEF = EPSI U5 MOM1 'LINE' ;

$$\boldsymbol{\varepsilon} = \frac{1}{2} \left(\mathbf{grad}(u) + \mathbf{grad}^T(u) \right)$$

$$\{\boldsymbol{\varepsilon}\} = [B]\{u\}$$

* CONTRAINTES A PARTIR DES DEFORMATIONS

SIG = ELAS DEF MOM1 MAM1 ;

$$\{\boldsymbol{\sigma}\} = [C]\{\boldsymbol{\varepsilon}\}$$

En un seul coup :

* CONTRAINTES A PARTIR DES DEPLACEMENTS

SIG = SIGM U5 MOM1 MAM1 'LINE' ;

$$\{\boldsymbol{\sigma}\} = [C][B]\{u\}$$

■ Post traitement : contraintes sur maillage déformé

* TRACE DES CONTRAINTES

TRAC (*A vous de jouer*) ;

■ Post traitement : contraintes sur maillage déformé


* TRACE DES CONTRAINTES

TRAC SIG MOM1 ;


* TRACE DES CONTRAINTES SUR CONFIGURATION DERFORMEE

DEF_5B = DEFO SU U5 150. ;

TRAC SIG MOM1 DEF_5B CSU ;



VONMISES
< 8.98E+07
> 3.01E+06



■ Post traitement : contraintes aux points de Gauss

* TRACE DES CONTRAINTES AUX POINTS DE GAUSS

SIGYY = EXCO SIG 'SMYY' ;





MESS 'Mini et Maxi de Sigma YY : ' (MINI SIGYY) (MAXI SIGYY) ;

TRAC SIGYY MOM1 ;

SIGYYG = CHAN 'CHPO' SIGYY MOM1 'SUPP' ;

MPGAUSS = EXTR SIGYYG 'MAIL' ;

TRAC SIGYYG MPGAUSS ;




■ Post traitement : évolution de la concentration de contrainte (le long du côté bas)

```

SIGB      = CHAN 'CHPO' SIG MOM1 ;
LBASG     = LBAS ELEM 'COMP' PB PA ;
EVSIG     = EVOL 'CHPO' SIGB 'SMYY' LBASG ;
EVK       = (ABS EVSIG) / 30.E6 ;
DESS EVK ;



```



■ Post traitement : évolution le long d'une ligne quelconque

```
OPTI 'ELEM' 'SEG2' ;
LIG1   = DROI 50 (0.1 0.1) (2.3 0.9) COUL 'ORAN' ;
TRAC (CSU ET LIG1) ;
```

```
SIGT   = CHAN 'NOEUD' MOM1 SIG ;
SIGLBAS = PROI SIGT LIG1 ;
EVSIG2 = EVOL 'ORAN' 'CHPO' SIGLBAS 'SMXY' LIG1 ;
DESS EVSIG2 ;
```



■ Post traitement : efforts de réaction aux appuis

* FORCES DE REACTION REPRESENTEES PAR DES FLECHES

REAC1 = REAC U5 (BLMX ET BLMY) ; (réaction due à un blocage)

VREAC = VECT REAC1 'FORC' 'ROUG' ; (flèches pour le tracé)

* IDEM POUR LES FORCES APPLIQUEES

VFIMP = (*A vous de jouer*)

TRAC (*A vous de jouer*) ;

■ Post traitement : efforts de réaction aux appuis

* FORCES DE REACTION REPRESENTEES PAR DES FLECHES


REAC1 = REAC U5 (BLMX ET BLMY) ; (réaction due à un blocage)

VREAC = VECT REAC1 'FORC' 'ROUG' ; (flèches pour le tracé)

* IDEM POUR LES FORCES APPLIQUEES

VFIMP = VECT FS 'FORC' 'VERT' ;

TRAC (VFIMP ET VREAC) CSU ;



■ Bonus : calcul en déplacements imposés

* MATRICES DE BLOCAGE

BLMY2 = BLOQ LHAUT 'UY' ;
RITOT2 = RI ET BLMX ET BLMY ET BLMY2 ;

[lien](#)

* SECOND MEMBRE ASSOCIE AU BLOCAGE EN DEPLACEMENT NON NUL


FU = DEPI BLMY2 -1.E-3 ;

* RESOLUTION

U52 = RESO RITOT2 (FU ET FV) ;


* POST TRAITEMENT

DEF_5C2 = DEFO CSU U52 150. 'VERT' ;
TRAC (DEF_INIC ET DEF_5C ET DEF_5C2) ;




Objectif : *calcul mécanique précédent*
+ *chargement thermique*

1. *calcul de la déformation thermique*
2. *ajout des forces nodales équivalentes à la déformation thermique* ([lien](#))



+



■ Forces nodales dues aux déformations thermiques (2nd membre)

* DEFORMATIONS THERMIQUES AU DERNIER CHAMP DE TEMPERATURE

* DU CALCUL THERMIQUE AVEC CONVECTION + RAYONNEMENT

TFINAL = TAB1 . 'TEMPERATURES' . (N1 - 1) ;

EPT = EPTH TFINAL MOM1 MAM1 ;

$$\{\varepsilon\}^{th} = [\alpha]\{T - \mathbf{TALP}\}$$

* PSEUDO CONTRAINTES POUR CES DEFORMATIONS THERMIQUES

SIT = ELAS EPT MOM1 MAM1 ;

$$\{\sigma\}^{th} = [C]\{\varepsilon\}^{th}$$

* FORCES NODALES POUR CETTE DEFORMATION THERMIQUE

FT = BSIG SIT MOM1 ;

$$\{F\}^{th} = \int_V [B]^T \{\sigma\}^{th} dV$$

■ Résolution du système linéaire (ajout d'un terme au 2nd membre)

* DEPLACEMENTS, PAR APPEL AU SOLVEUR EN SUPERPOSANT LES FORCES

* PUREMENT MECANIQUES ET LES PSEUDO FORCES THERMIQUES

U6 = (A vous de jouer);

■ Forces nodales dues aux déformations thermiques (2nd membre)

* DEFORMATIONS THERMIQUES AU DERNIER CHAMP DE TEMPERATURE

* DU CALCUL THERMIQUE AVEC CONVECTION + RAYONNEMENT

TFINAL = TAB1 . 'TEMPERATURES' . (N1 - 1) ;

EPT = EPTH TFINAL MOM1 MAM1 ;

$$\{\varepsilon\}^{th} = [\alpha]\{T - \mathbf{TALP}\}$$

* PSEUDO CONTRAINTES POUR CES DEFORMATIONS THERMIQUES

SIT = ELAS EPT MOM1 MAM1 ;

$$\{\sigma\}^{th} = [C]\{\varepsilon\}^{th}$$

* FORCES NODALES POUR CETTE DEFORMATION THERMIQUE

FT = BSIG SIT MOM1 ;

$$\{F\}^{th} = \int_V [B]^T \{\sigma\}^{th} dV$$

■ Résolution du système linéaire (ajout d'un terme au 2nd membre)

* DEPLACEMENTS, PAR APPEL AU SOLVEUR EN SUPERPOSANT LES FORCES

* PUREMENT MECANIQUES ET LES PSEUDO FORCES THERMIQUES

U6 = RESO RITOT (FS ET FV ET FT) ;

■ Post traitement : maillage déformé, déformations, contraintes

* DEFORMEE THERMOMECANIQUE

DEF_6 = (A vous de jouer)

DEF_6C = (A vous de jouer)

TRAC (A vous de jouer) ;


■ Post traitement : maillage déformé, déformations, contraintes

* DEFORMEE THERMOMECANIQUE

DEF_6 = DEFO SU U6 150. ;

DEF_6C = DEFO CSU U6 150. 'ORAN' ;

TRAC (DEF_INIC ET DEF_5C ET DEF_6C) ;



■ Post traitement : maillage déformé, déformations, contraintes

* DEFORMATIONS TOTALES

EP = (A vous de jouer, opérateur EPSI)

* DEFORMATIONS ELASTIQUES

EPE = (A vous de jouer)

* LES CONTRAINTES SONT CALCULEES A PARTIR DES DEFORMATIONS ELASTIQUES

SIGT = (A vous de jouer)

TRAC SIGT MOM1 DEF_6 CSU ;

■ Post traitement : maillage déformé, déformations, contraintes

* DEFORMATIONS TOTALES

EP = EPSI MOM1 U6 'LINE' ;


* DEFORMATIONS ELASTIQUES

EPE = EP - EPT ;

* LES CONTRAINTES SONT CALCULEES A PARTIR DES DEFORMATIONS ELASTIQUES

SIGT = ELAS EPE MOM1 MAM1 ;


TRAC SIGT MOM1 DEF_6 CSU ;



Objectif : calcul thermo-mécanique précédent
+ caractéristique α variable dans l'espace

1. calcul du champ $\alpha(x)$ (à partir des coordonnées)
2. caractéristique matériau décrite par ce champ
3. mise à jour des forces nodales de déformation thermique

$$\alpha(x) = \alpha_0 \left(1 + \beta \left(\frac{x - x_{\text{moy}}}{x_{\text{max}} - x_{\text{min}}} \right)^3 \right)$$



■ Coefficient de dilatation thermique variable dans l'espace

XX = COOR SU 1 ;

XMAX = MAXI XX ;

XMIN = MINI XX ;

XMOY = 0.5 * (XMAX + XMIN) ;

BETA = 7. ;

CHP_ALPH = ALPHAMAT * (1. + (BETA * (((XX - XMOY) / (XMAX - XMIN)) ** 3))) ;

■ Mise à jour des caractéristiques du matériau

* CONVERSION DU CHPOINT -> EN MCHAML

CHM_ALPH = CHAN 'CHAM' CHP_ALPH MOM1 ;

* MISE A JOUR DU MCHAML DU MATERIAU AVEC ALPHA VARIABLE

MAM1B = MATE MOM1 'YOUN' YOUNGMAT 'NU' NUMAT
'ALPH' CHM_ALPH 'TREF' T0 'TALP' T0 ;

■ Mise à jour des forces nodales dues aux déformations thermiques (2nd membre)

* MISE A JOUR DES DEFORMATIONS THERMIQUES PURES AVEC ALPHA VARIABLE
 $EPT = (A \text{ vous de jouer, opérateur } EPTH)$

* MISE A JOUR DES PSEUDO CONTRAINTES THERMIQUES AVEC ALPHA VARIABLE
 $SIT = (A \text{ vous de jouer, opérateur } ELAS)$

* MISE A JOUR DES FORCES NODALES POUR CETTE DEFORMATION THERMIQUE
 $FT = (A \text{ vous de jouer, opérateur } BSIG)$

■ Résolution du système linéaire

* DEPLACEMENTS AVEC ALPHA VARIABLE
 $U7 = \text{RESO RITOT (FS ET FV ET FT) ;}$

■ Mise à jour des forces nodales dues aux déformations thermiques (2nd membre)

* MISE A JOUR DES DEFORMATIONS THERMIQUES PURES AVEC ALPHA VARIABLE
EPT = EPTH TFINAL MOM1 MAM1B ;

* MISE A JOUR DES PSEUDO CONTRAINTES THERMIQUES AVEC ALPHA VARIABLE
SIT = ELAS EPT MOM1 MAM1B ;

* MISE A JOUR DES FORCES NODALES POUR CETTE DEFORMATION THERMIQUE
FT = BSIG SIT MOM1 ;

■ Résolution du système linéaire

* DEPLACEMENTS AVEC ALPHA VARIABLE
U7 = RESO RITOT (FS ET FV ET FT) ;

■ Post traitement : maillage déformé, déformations, contraintes

* DEFORMEE AVEC ALPHA VARIABLE

(A vous de jouer)

(A vous de jouer)

TRAC *(A vous de jouer)* ;


■ Post traitement : maillage déformé, déformations, contraintes

* DEFORMEE AVEC ALPHA VARIABLE

```
DEF_7 = DEFO SU U7 150. ;
```

```
DEF_7C = DEFO CSU U7 150. 'ROUG' ;
```

```
TRAC (DEF_INIC ET DEF_6C ET DEF_7C) ;
```



■ Post traitement : maillage déformé, déformations, contraintes

* DEFORMATIONS TOTALES AVEC ALPHA VARIABLE

EP = EPSI U7 MOM1 'LINE' ;


* DEFORMATION ELASTIQUES AVEC ALPHA VARIABLE

EPE = EP - EPT ;

* CONTRAINTES AVEC ALPHA VARIABLE


SIGT = ELAS EPE MOM1 MAM1B ;

TRAC SIGT MOM1 DEF_7 CSU ;



Objectif : *calcul thermique transitoire + mécanique*
+ *matériau élasto-plastique parfait*

1. reprendre le calcul thermique transitoire précédent
2. modèle mécanique avec comportement non linéaire
3. description temporelle des chargements mécaniques
4. résolution avec la procédure PASAPAS



■ Mise à jour du modèle et des caractéristiques du matériau

* MISE A JOUR DU MODELE MECANIQUE

MOM2 = MODE SU 'MECANIQUE' 'ELASTIQUE' 'PLASTIQUE' 'PARFAIT' ;

* MISE A JOUR DU MATERIAU (AJOUT DE LA LIMITE D'ELASTICITE 'SIGY')

MAM2 = MATE MOM2 'YOUN' YOUNGMAT 'NU' NUMAT 'SIGY' SIGYMAT
'ALPH' CHM_ALPH 'TREF' T0 'TALP' T0 ;

■ Chargements mécaniques

* EVOLUTIONS TEMPORELLES DES CHARGEMENTS MECANIQUES

EVTM = EVOL 'MANU' (PROG 0. (0.98 * TPSFIN) TPSFIN)
(PROG 0. 0. 1.) ;

CHAFS = CHAR 'MECA' FS EVTMT ;

CHAFV = CHAR 'MECA' FV EVCTE ;

CHAM = CHAFS ET CHAFV ;

■ Construction de la table pour la procédure PASAPAS

* PREPARATION DE LA TABLE POUR PASAPAS

TAB2 = TABL ;

.....

.....

.....

.....

(A vous de jouer)

.....

.....

.....

PASAPAS TAB2 ;

■ Construction de la table pour la procédure PASAPAS




* PREPARATION DE LA TABLE POUR PASAPAS

```
TAB2                                = TABL ;
TAB2 . 'MODELE'                      = MOT ET MOC ET MOR ET MOM2 ;
TAB2 . 'CARACTERISTIQUES'           = MAT ET MAC ET MAR ET MAM2 ;
TAB2 . 'BLOCAGES_MECANIQUES'        = BLMX ET BLMY ;
TAB2 . 'BLOCAGES_THERMIQUES'        = BLT ;
TAB2 . 'CHARGEMENT'                  = CHAT ET CHAM ;
TAB2 . 'TEMPS_CALCULES'              = PROG 0. 'PAS' (0.02 * TPSFIN) (0.98 * TPSFIN)
                                     'PAS' (0.001 * TPSFIN) TPSFIN ;
TAB2 . 'CELSIUS'                     = VRAI ;
PASAPAS TAB2 ;
```

■ Post traitement : contraintes


* BOUCLE SUR LES PAS DE TEMPS

```
LISO2 = PROG 0. 'PAS' 10.E6 100.E6 ;
N2 = DIME (TAB2 . 'DEPLACEMENTS') ;
REPE B1 N2 ;
  U_I = TAB2 . 'DEPLACEMENTS' . (&B1 - 1) ;
  DEF_I = DEFO SU U_I 150. ;
  S_I = TAB2 . 'CONTRAINTES' . (&B1 - 1) ;
  TRAC S_I MOM2 DEF_I CSU LIS02 ;
FIN B1 ;
DEF_8C = DEFO CSU U_I 150. 'VIOL' ;
```




■ Post traitement : déformations plastiques cumulées [\(lien\)](#)

```
* BOUCLE SUR LES PAS DE TEMPS
VI      = TAB2 . 'VARIABLES_INTERNES' . (N2 - 1) ;
EQ_MAX  = MAXI (EXCO 'EPSE' VI) ;
LISO3   = PROG 0. 'PAS' (EQ_MAX / 10.) EQ_MAX ;
REPE B1 N2 ;
  U_I    = TAB2 . 'DEPLACEMENTS' . (&B1 - 1) ;
  DEF_I  = DEFO SU U_I 150. ;
  V_I    = TAB2 . 'VARIABLES_INTERNES' . (&B1 - 1) ;
  TRAC V_I MOM2 DEF_I CSU LISO3 ;
FIN B1 ;
```



Objectif : calcul thermo-mécanique précédent
+ conductivité *dépendante de la température*

1. caractéristique matériau mécanique variable et dépendant d'une inconnue du calcul thermique



■ Mise à jour des caractéristiques du matériau

* VARIATION DE LA CONDUCTIVITE EN FONCTION DE T

EVLAM = EVOL 'MANU' 'T' (PROG 50. 200.) 'K' (PROG 210. 250.) ;

* MISE A JOUR DU MATERIAU

MAT2 = MATE MOT 'K' EVLAM 'C' CAPAMAT 'RHO' RHOMAT ;

■ Puis on relance PASAPAS

(A vous de jouer)

■ Mise à jour des caractéristiques du matériau

* VARIATION DE LA CONDUCTIVITE EN FONCTION DE T

EVLAM = EVOL 'MANU' 'T' (PROG 50. 200.) 'K' (PROG 210. 250.) ;

* MISE A JOUR DU MATERIAU

MAT2 = MATE MOT 'K' EVLAM 'C' CAPAMAT 'RHO' RHOMAT ;

■ Puis on relance PASAPAS


```
TAB2 = TABL ;
TAB2 . 'MODELE' = MOT ET MOC ET MOR ET MOM2 ;
TAB2 . 'CARACTERISTIQUES' = MAT2 ET MAC ET MAR ET MAM2 ;
TAB2 . 'BLOCAGES_MECANIQUES' = BLMX ET BLMY ;
TAB2 . 'BLOCAGES_THERMIQUES' = BLT ;
TAB2 . 'CHARGEMENT' = CHAT ET CHAM ;
TAB2 . 'TEMPS_CALCULES' = PROG 0. 'PAS' (0.02 * TPSFIN) (0.98 * TPSFIN)
                          'PAS' (0.001 * TPSFIN) TPSFIN ;
TAB2 . 'CELSIUS' = VRAI ;
PASAPAS TAB2 ;
```

■ Post traitement : maillage déformé

* COMPARAISON DE LA DEFORMEE FINALE AVEC LES CALCULS PRECEDENTS

```

N2      = DIME (TAB2 . 'DEPLACEMENTS') ;
U9      = TAB2 . 'DEPLACEMENTS' . (N2 - 1) ;
DEF_9   = DEFO SU U9 150. ;
DEF_9C  = DEFO CSU U9 150. 'BLEU' ;
TRAC (DEF_INIC ET DEF_7C ET DEF_8C ET DEF_9C) ;
  
```




■ Post traitement : déformation plastique cumulée

```

VI9      = TAB2 . 'VARIABLES_INTERNES' . (N2 - 1) ;
EQ_MAX   = MAXI (EXCO 'EPSE' VI9) ;
LIS03    = PROG 0. 'PAS' (EQ_MAX / 10.) EQ_MAX ;
U9       = TAB2 . 'DEPLACEMENTS' . (N2 - 1) ;
DEF9     = DEFO SU U9 150. ;
TRAC    VI9 MOM2 DEF9 CSU LIS03 ;
  
```

EPSE
< 1.46E-03
> -2.30E-05




1.40E-03
1.26E-03
1.12E-03
9.80E-04
8.40E-04
7.00E-04
5.60E-04
4.20E-04
2.80E-04
1.40E-04
0.0

COMPLÉMENTS

- Le choix des éléments finis se fait dans MODE

```
MODP1 = MODE LIG1 'MECANIQUE' 'ELASTIQUE' | 'POUT' | ;
                                             | 'TIMO' |
```

```
MODC1 = MODE SUR1 'MECANIQUE' 'ELASTIQUE' | 'DKT' | ;
                                             | 'COQ4' |
```



- Il convient alors de fournir les caractéristiques géométriques

```
MATP1 = MATE MODP1 'YOUN' 210.E9 'NU' 0.3
          'SECT' 1.E-2 'INRY' 1.E-4 'INRZ' 2.E-4 'TORS' 3.E-4 ;
```

```
MATC1 = MATE MODC1 'YOUN' 210.E9 'NU' 0.3
          'EPAI' 1.E-2 ;
```

- On peut alors agir sur les d.d.l. de déplacement et de rotation

```
BL1 = BLOQ P1 'UX' 'UY' 'UZ' 'RX' 'RY' 'RZ' ;
```

■ Exemple d'une coque cylindrique sous pression (en 3D)

```


OPTI 'DIME' 3 'ELEM' 'QUA4' ;
p1 = 4. 0. 0. ;
lig1 = CERC 45 'ROTA' 360. p1 (0. 0. 0.) (0. 0. 1.) ;
ELIM lig1 1.E-2 ;
sur1 = lig1 TRAN 20 (0. 0. 10.) ;

mo = MODE sur1 'MECANIQUE' 'COQ4' ;
ma = MATE mo 'YOUN' 210.E9 'NU' 0.3 'EPAI' 0.3 ;
rig = RIGI mo ma ;

bl1 = BLOQ 'UX' 'UY' 'UZ' 'RX' 'RY' 'RZ' lig1 ;
f1 = PRES 'COQU' mo 42.E5 'NORM' ma ;

u = RESO (rig ET bl1) f1 ;
def1 = DEFO sur1 u ;
TRAC 'CACH' def1 ;

```



■ Le même cas (en 2D axisymétrique)

```
OPTI 'DIME' 2 'MODE' 'AXIS' 'ELEM' 'SEG2' ;
```

```
p1 = 4. 0. ;
```

```
lig1 = DROI 20 p1 (4. 10.) ;
```

```
mo = MODE lig1 'MECANIQUE' 'COQ2' ;
```

```
ma = MATE mo 'YOUN' 210.E9 'NU' 0.3 'EPAI' 0.3 ;
```

```
rig = RIGI mo ma ;
```

```
b11 = BLOQ 'UR' 'UZ' 'RT' p1 ;
```

```
f1 = PRES 'COQU' mo -42.E5 'NORM' ;
```

```
u = RESO (rig ET b11) f1 ;
```

```
def1 = DEFO lig1 u ;
```

```
TRAC def1 ;
```

■ Exemple d'un problème de conduction stationnaire axisymétrique


```
OPTI 'DIME' 2 'MODE' 'AXIS' 'ELEM' 'QUA8' ;
```

```
p1 = 10. 0. ;
p2 = 50. 0. ;
lig1 = DROI 10 p1 p2 ;
sur1 = lig1 TRAN 1 (0. 1.) ;
cot1 = sur1 COTE 4 ;
cot2 = sur1 COTE 2 ;
```

```
mo = MODE sur1 'THERMIQUE' ;
ma = MATE mo 'K' 42. ;
con = COND mo ma ;
```

```
b11 = BLOQ 'T' cot1 ;
b12 = BLOQ 'T' cot2 ;
f1 = DEPI b11 20. ;
f2 = DEPI b12 500. ;
```

```
t = RESO (con ET b11 ET b12) (f1 ET f2) ;
TRAC t sur1 ;
evt = EVOL 'CHPO' t 'T' lig1 ;
DESS evt ;
```



■ Le même cas en 1D (cylindrique)

```
OPTI 'DIME' 1 'MODE' 'UNID' 'AXIS' 'ELEM' 'SEG3' ;
```

```
p1 = POIN 10. ;
```

```
p2 = POIN 50. ;
```

```
lig1 = DROI 10 p1 p2 ;
```

```
mo = MODE lig1 'THERMIQUE' ;
```

```
ma = MATE mo 'K' 42. ;
```

```
con = COND mo ma ;
```

```
b11 = BLOQ 'T' p1 ;
```

```
b12 = BLOQ 'T' p2 ;
```

```
f1 = DEPI b11 20. ;
```


```
f2 = DEPI b12 500. ;
```

```
t = RESO (con ET b11 ET b12) (f1 ET f2) ;
```

```
TRAC t lig1 ;
```

```
evt = EVOL 'CHPO' t 'T' lig1 ;
```

```
DESS evt ;
```



■ Au format binaire : sauvegarde / restitution

```
OPTI 'SAUV' 'toto' ; SAUV ;
OPTI 'REST' 'toto' ; REST ;
```

Mais aussi possible au format texte (lourd !)

■ Exécuter une commande EXTErieure

```
TAB1 = EXTE 'grep -in mecanique toto.dgibi' 'RC' ;
→ TAB1 contient le résultat de la commande grep
```

■ Directive ACQUérir

Acquérir dans un fichier texte, ligne par ligne

```
OPTI 'ACQU' 'fichier.txt' ;
ACQU N1*'ENTIER' A*'FLOTTANT' ;
ACQU N2*'ENTIER' L1*'LISTREEL' 4 ;
```

fichier.txt

1	3.14	X	Y	Z
2	25.2	28.3	24.3	16.6

■ Écrire dans un fichier texte → en utilisant la directive **SORT**

```

OPTI 'SORT' 'mon_fichier.txt' ;

SORT 'CHAI' 'J''ecris dans un fichier !' ;
SORT 'CHAI' 4 8 15 16 23 42 ;
SORT 'CHAI' ' ' ;

SORT 'CHAI' 'N° iteration      Fibonacci' ;
FIBO1 = 0 ;
FIBO2 = 1 ;
SORT 'CHAI' 0      FIBO1 ;
SORT 'CHAI' 1      FIBO2 ;
REPE B1 15 ;
    FIBO2B = FIBO2 ;
    FIBO2 = FIBO1 + FIBO2 ;
    FIBO1 = FIBO2B ;
    SORT 'CHAI' (&B1 + 1)  FIBO2 ;
FIN  B1 ;

```

mon_fichier.txt

```

J'ecris dans un fichier !
  4      8      15      16      23      42

N° iteration      Fibonacci
  0              0
  1              1
  2              1
  3              2
  4              3
  5              5
  6              8
  7             13
  8             21
  9             34
 10            55
 11            89
 12           144
 13           233
 14           377
 15           610
 16           987

```

■ **LIRE / SORTIR** lecture / écriture sous différents formats

Lire/écrire au format texte tabulé (CSV ou autres)

objets concernés : listes, évolutions, TABLE
utilisé par tout éditeur de texte ou tableur (Excel)



```
OPTI 'SORT' 'fichier.csv' ;  
SORT 'EXCE' OBJET1 ;
```

```
TAB1 = 'LIRE' 'CSV' 'fichier.csv' ;
```

Choix possible du séparateur de colonnes :
point virgule, virgule, espace, tabulation, barre oblique

■ LIRE / SORTir lecture / écriture sous différents formats

Lire au format UNV (fichier texte)

objets concernés : MAILLAGE

utilisé par Gmsh, Salomé, HyperMesh, ...



```
TAB1 = LIRE 'UNV' 'fichier.unv' ;
```

Lire/écrire au format AVS (fichier texte)

objets concernés : MAILLAGE, CHPOINT, MCHAML

extension .inp utilisé par Abaqus



```
OPTI 'SORT' 'fichier.inp' ;
SORT 'AVS' OBJET1 'TEMPS' 12.3 ;
```

```
OPTI 'LECT' 'fichier.inp' ;
TAB1 = LIRE 'AVS' ;
```

■ LIRE / SORTir lecture / écriture sous différents formats

Écrire au format VTK

objets concernés : MAILLAGE, CHPOINT, MCHAML
utilisé par Paraview



```
OPTI 'SORT' 'fichier.vtk' ;  
SORT 'VTK' MAIL1 'VIS' MAIL2 'ECROU'  
DEP1 'DEPLACEMENTS' SIG1 'CONTRAINTES' ;
```

■ **LIRE / SORTir** lecture / écriture sous différents formats

Lire/écrire au format MED

objets concernés : MAILLAGE, CHPOINT, TABLE (issue de PASAPAS)
utilisé par Salomé, Europlexus



```
OPTI 'SORT' 'fichier.med' ;  
SORT 'MED' OBJET1 ;
```

```
TAB1 = LIRE 'MED' 'fichier.med' ;
```

Lire au format FEM

objets concernés : MAILLAGE
utilisé par HyperMesh (Altair)



```
TAB1 = LIRE 'FEM' 'fichier.fem' ;
```

- Écrire vos procédures dans des fichiers textes avec l'extension ".procedur"
- Lancer la commande castem22 avec l'option –u
castem22 -u
 - ➔ création d'un fichier **UTILPROC** (fichier d'accès direct)
- On peut alors lancer Cast3M et toutes les procédures seront disponibles
Le fichier **UTILPROC** doit être présent dans le répertoire de travail

Idem pour les notices (fichiers avec extension ".notice")

L'utilisateur peut **modifier/corriger/ajouter** le code source des opérateurs et directives

■ Compilation des fichiers source Esope

```
compilcast22 toto.eso tata.eso  
compilcast22 tata.eso
```

...

■ Édition des liens

```
essaicast22
```

→ création d'un fichier exécutable binaire : **cast_64_22**

→ version locale de Cast3M

■ Se lancer comme d'habitude

```
castem22 mon_fichier.dgibi
```

- **Consulter la documentation régulièrement**
~70 instructions découvertes durant cette formation
près de 1400 instructions existantes !
- **Inscription à la liste de diffusion Cast3M (voir le site web Cast3M)**
Envoyer un e-mail vide à sympa@umontpellier.fr avec
comme objet du message :
SUB cast3m-util Votre_nom Votre_prenom

et rien d'autre ! (pas de message, pas de signature, ...)
- **Club Cast3M : séminaire annuel des utilisateurs**
Chaque année en novembre dans le sud de Paris
Présentation de travaux réalisés avec Cast3M, nouveautés de la
prochaine version
Inscription gratuite !

THERMO-ÉLASTICITÉ LINÉAIRE

(1) RAPPEL DES ÉQUATIONS

Déformations totales (linéarisés)

$$\boldsymbol{\varepsilon} = \frac{1}{2} \left(\mathbf{grad}(u) + \mathbf{grad}^T(u) \right) \quad \varepsilon_{ij} = \frac{1}{2} \left(\frac{\partial u_i}{\partial x_j} + \frac{\partial u_j}{\partial x_i} \right)$$

Partition des déformations

$$\boldsymbol{\varepsilon} = \boldsymbol{\varepsilon}^e + \boldsymbol{\varepsilon}^{th}$$

Déformations thermiques

$$\boldsymbol{\varepsilon}^{th} = \alpha \Delta T \boldsymbol{\delta}$$

Élasticité

$$\boldsymbol{\sigma} = \mathbf{C} : \boldsymbol{\varepsilon}^e = \mathbf{C} : (\boldsymbol{\varepsilon} - \boldsymbol{\varepsilon}^{th})$$

avec :	u	déplacement		
	$\boldsymbol{\varepsilon}$	déformation totale	$\boldsymbol{\sigma}$	contrainte
	$\boldsymbol{\varepsilon}^e$	déformation élastique	$\boldsymbol{\varepsilon}^{th}$	déformation thermique
	$\boldsymbol{\delta}$	tenseur identité		
	\mathbf{C}	tenseur de rigidité (ordre 4)		
	α	dilatation thermique		
	ΔT	élévation de température		

THERMO-ÉLASTICITÉ LINÉAIRE (2) DÉFORMATIONS PLANES

Hypothèse déformations planes :

$$\boldsymbol{\sigma} = \begin{bmatrix} \sigma_{11} & \sigma_{12} & 0 \\ \sigma_{12} & \sigma_{22} & 0 \\ 0 & 0 & \sigma_{33} \end{bmatrix} \quad \boldsymbol{\varepsilon} = \begin{bmatrix} \varepsilon_{11} & \varepsilon_{12} & 0 \\ \varepsilon_{12} & \varepsilon_{22} & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

Notation vectorielle :

$$\{\boldsymbol{\sigma}\} = \begin{Bmatrix} \sigma_{11} \\ \sigma_{22} \\ \sigma_{12} \end{Bmatrix} \quad \{\boldsymbol{\varepsilon}\} = \begin{Bmatrix} \varepsilon_{11} \\ \varepsilon_{22} \\ 2\varepsilon_{12} \end{Bmatrix}$$

avec : $\sigma_{33} = \nu(\sigma_{11} + \sigma_{22})$
3 composantes indépendantes

Loi de Hooke :

$$\begin{Bmatrix} \sigma_{11} \\ \sigma_{22} \\ \sigma_{12} \end{Bmatrix} = \underbrace{\frac{E}{1-\nu^2} \begin{bmatrix} 1 & \nu & 0 \\ \nu & 1 & 0 \\ 0 & 0 & \frac{1-\nu}{2} \end{bmatrix}}_{[\mathbf{C}]} \begin{Bmatrix} \varepsilon_{11} \\ \varepsilon_{22} \\ 2\varepsilon_{12} \end{Bmatrix}$$

$$\{\boldsymbol{\sigma}\} = [\mathbf{C}]\{\boldsymbol{\varepsilon}\}$$

$[\mathbf{C}]$ matrice de Hooke


THERMO-ÉLASTICITÉ LINÉAIRE PLANE

(3) ÉLÉMENT FINI

Interpolation des inconnues primales :

$$\{u(x)\} = [N(x)]\{q\}$$

en tout point (x, y) de l'élément
 $\{q\}$ sont les déplacements des nœuds de l'élément



Cas d'un élément TRI3 (6 ddl) :

$$\begin{Bmatrix} u_x(x) \\ u_y(x) \end{Bmatrix} = \begin{bmatrix} N_A(x) & 0 & N_B(x) & 0 & N_C(x) & 0 \\ 0 & N_A(x) & 0 & N_B(x) & 0 & N_C(x) \end{bmatrix} \begin{Bmatrix} q_1^A \\ q_2^A \\ q_1^B \\ q_2^B \\ q_1^C \\ q_2^C \end{Bmatrix}$$

Déformations (linéarisées) :

$$\{\varepsilon(x)\} = \begin{bmatrix} N_{A,x_1} & 0 & N_{B,x_1} & 0 & N_{C,x_1} & 0 \\ 0 & N_{A,x_2} & 0 & N_{B,x_2} & 0 & N_{C,x_2} \\ N_{A,x_1} & N_{A,x_2} & N_{B,x_1} & N_{B,x_2} & N_{C,x_1} & N_{C,x_2} \end{bmatrix} \{q\} \quad \text{avec } N_{A,x_i} = \frac{\partial N_A(x)}{\partial x_i}$$

$$\{\varepsilon(x)\} = [B(x)]\{q\}$$

$[B(x)]$ opérateur gradient discrétisé

Énergie de déformation élastique élémentaire :

$$\begin{aligned}
 e_{def} &= \frac{1}{2} \int_{V_e} \boldsymbol{\sigma} : \boldsymbol{\varepsilon} \, dV \\
 &= \frac{1}{2} \int_{V_e} \{\boldsymbol{\varepsilon}\}^T \{\boldsymbol{\sigma}\} \, dV \\
 &= \frac{1}{2} \int_{V_e} \{q\}^T [B]^T [C] [B] \{q\} \, dV \\
 &= \frac{1}{2} \{q\}^T \underbrace{\left[\int_{V_e} [B]^T [C] [B] \, dV \right]}_{[k_e]} \{q\}
 \end{aligned}$$

V_e "volume" de l'élément fini
 $[k_e]$ matrice de rigidité élémentaire
 (calculé par intégration numérique)

Assemblage :

$$E_{def} = \frac{1}{2} \{Q\}^T \sum_e [k_e] \{Q\} = \frac{1}{2} \{Q\}^T [K] \{Q\}$$

$\{Q\}$ déplacements des nœuds du maillage
 Les matrices $[k_e]$ sont étendues sur tous les ddl du maillage avec des termes nuls
 $[K]$ matrice de rigidité assemblée

THERMO-ÉLASTICITÉ LINÉAIRE

(5) PRINCIPE TRAVAUX VIRTUELS

Principe des travaux virtuels :

f_v densité volumique d'efforts imposés
 f_s densité surfaciques d'efforts imposés

$$\int_{V_e} \boldsymbol{\sigma} : \boldsymbol{\delta \varepsilon}^* dV = \int_{V_e} f_v \delta u^* dV + \int_{S_e} f_s \delta u^* dS$$

$$\int_{V_e} \{\delta \varepsilon^*\}^T \{\boldsymbol{\sigma}\} dV = \int_{V_e} \{\delta u^*\}^T \{f_v\} dV + \int_{S_e} \{\delta u^*\}^T \{f_s\} dS$$

$$\{\delta q^*\}^T \int_{V_e} [B]^T \{\boldsymbol{\sigma}\} dV = \{\delta q^*\}^T \left(\int_{V_e} [N]^T \{f_v\} dV + \int_{S_e} [N]^T \{f_s\} dS \right)$$

$$\underbrace{\int_{V_e} [B]^T [C] [B] dV}_{[k_e]} \{q\} = \underbrace{\int_{V_e} [N]^T \{f_v\} dV + \int_{S_e} [N]^T \{f_s\} dS}_{\{F_e\}}$$

$$[k_e] \{q\} = \{F_e\}$$

$\{F_e\}$ forces nodales équivalentes

On ajoute un terme au second membre :

$$\int_{V_e} [B]^T \{\sigma\} dV = \{F_e\}$$

$$\int_{V_e} [B]^T [C] \{\varepsilon - \varepsilon^{th}\} dV = \{F_e\}$$

$$\int_{V_e} [B]^T [C] \{\varepsilon\} dV = \{F_e\} + \underbrace{\int_{V_e} [B]^T [C] \{\varepsilon^{th}\} dV}_{\{F_{th}\}}$$

$$[k_e] \{q\} = \{F_e\} + \{F_{th}\}$$

- En mécanique on résout le problème contraint suivant :

$$\begin{cases} [K]\{Q\} = \{F\} \\ [b]\{Q\} = \{Q_0\} \end{cases}$$

$[b]$ matrice de blocage
 $\{Q_0\}$ valeurs imposées

- La fonction à minimiser est :

$$f\{Q\} = \frac{1}{2}\{Q\}^T \cdot [K] \cdot \{Q\} - \{Q\}^T \cdot \{F\}$$

avec les contraintes : $[b]\{Q\} - \{Q_0\} = \{0\}$

- Le théorème de Lagrange nous dit qu'il existe des réels $\{\lambda\}$ tels que :

$$[K]\{Q\} - \{F\} + [b]^T \{\lambda\} = \{0\}$$

- On se ramène alors au système suivant :

$$\begin{bmatrix} K & b^T \\ b & 0 \end{bmatrix} \begin{Bmatrix} Q \\ \lambda \end{Bmatrix} = \begin{Bmatrix} F \\ Q_0 \end{Bmatrix}$$

■ Exemple simple : sur un élément barre

```
OPTI 'DIME' 2 'ELEM' 'SEG2' ;
```

```
* Maillage
```

```
p1 = 0. 0. ;
```

```
p2 = 3. 0. ;
```

```
l1 = DROI 1 p1 p2 ;
```

```
* Modele et parametres
```

```
mo = MODE l1 'MECANIQUE' 'BARR' ;
```

```
ma = MATE mo 'YOUN' 210.E9 'NU' 0.3  
      'SECT' 1.E-2 ;
```

```
* Matrice de rigidite principale
```

```
rig1 = RIGI mo ma ;
```

```
LIST rig1 ;
```

```
* Un petit blocage
```

```
b11 = BLOQ 'UX' p1 ;
```

```
LIST b11 ;
```

```
* Et un second membre associe
```

```
f1 = DEPI b11 1.2 ;
```

```
LIST f1 ;
```

matrices

$$\text{rig1} = \begin{bmatrix} k & 0 & -k & 0 \\ 0 & 0 & 0 & 0 \\ -k & 0 & k & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

$$\text{b11} = \begin{bmatrix} 0 & 1 \\ 1 & 0 \end{bmatrix}$$

$$\text{rig1 ET b11} = \begin{bmatrix} k + 0 & 0 & -k & 0 & | & 1 \\ 0 & 0 & 0 & 0 & | & 0 \\ -k & 0 & k & 0 & | & 0 \\ 0 & 0 & 0 & 0 & | & 0 \\ \hline 1 & 0 & 0 & 0 & | & 0 \end{bmatrix}$$

inconnues primales

$$\begin{bmatrix} UX_1 \\ UY_1 \\ UX_2 \\ UY_2 \end{bmatrix}$$

$$\begin{bmatrix} LX_3 \\ UX_1 \end{bmatrix}$$

$$\begin{bmatrix} UX_1 \\ UY_1 \\ UX_2 \\ UY_2 \\ \hline LX_3 \end{bmatrix}$$

inconnues duales

$$\begin{bmatrix} FX_1 \\ FY_1 \\ FX_2 \\ FY_2 \end{bmatrix}$$

$$\begin{bmatrix} FLX_3 \\ FX_1 \end{bmatrix}$$

$$\begin{bmatrix} FX_1 \\ FY_1 \\ FX_2 \\ FY_2 \\ \hline FLX_3 \end{bmatrix}$$


■ Définition

$$p(t) = \int_0^t \dot{p}(\tau) d\tau \quad \dot{p} = \sqrt{\frac{2}{3} \dot{\boldsymbol{\varepsilon}}^p : \dot{\boldsymbol{\varepsilon}}^p} = \sqrt{\frac{2}{3} \dot{\varepsilon}^p_{ij} \dot{\varepsilon}^p_{ij}}$$

■ Signification

C'est une mesure de la **longueur du trajet de l'écoulement** dans l'espace des déformations plastiques

[\(lien\)](#)



DESCRIPTION DES OBJETS GIBIANE

■ Objets généraux

ENTIER	Nombre entier
FLOTTANT	Nombre réel
MOT	Chaine de caractères
LOGIQUE	Booléen (VRAI ou FAUX)
LISTENTI	Liste d'entiers
LISTREEL	Liste de réels
LISTMOT	Liste de mots (limités à 4 caractères)
TABLE	Ensemble d'objets de type quelconque et caractérisés par un indice de type quelconque
EVOLUTIO	Représentation d'un graphe d'une fonction, suite de couples $(x ; f(x))$

DESCRIPTION DES OBJETS

■ Objets pour le maillage

POINT Coordonnées d'un point de l'espace + densité

MAILLAGE Domaine de l'espace discrétisé

■ Objets pour le calcul

CHPOINT N'importe quel type de données aux nœuds d'un maillage (flottants, logiques, champs, ...)

MMODEL Association d'un maillage, d'une formulation EF et d'un comportement matériau
Définit les inconnues physiques primales / duales
ex : déplacements / forces
température / flux

DESCRIPTION DES OBJETS

■ Objets pour le calcul (suite)

MCHAML	N'importe quel type de données à l'intérieur des éléments d'un maillage (flottants, logiques, champs, ...) Variétés des points supports (points de Gauss, centre de gravité, nœuds, ...)
RIGIDITE	Matrice de raideur, de masse, de conductivité, ... Matrice couplant les inconnues physiques
CHARGEME	Description spatio-temporelle d'un chargement

■ Objets pour le post traitement

VECTEUR	Pour visualiser plusieurs composantes d'un champ
DEFORMEE	Pour visualiser un maillage déformé

François DI PAOLA
Commissariat à l'énergie atomique et aux énergies alternatives
Centre de Saclay | 91191 Gif-sur-Yvette Cedex

DES/DANS
DM2S
SEMT